<cfsetting enablecfoutputonly="Yes">

<!--- 
  /**
   * Part of the standard files that come along with a Mireau application. This
   * customtag will draw a border around a piece of content.
   *
   * @author  rens@joxx.nl
   * @version 1.0 - 2003/12/12
   * @version 1.1 - 2004/01/15  Made this customtag XHTML approved. (for the
   *                            'normal' mode anyway - the 'simple' mode has
   *                            not been tested yet)
   */
--->

<cfscript>
  if( not isDefined( "attributes.sBackgroundImg" ))     { attributes.sBackgroundImg     = request.site.s_NOT_SET; }
  if( not isDefined( "attributes.sBGColor" ))           { attributes.sBGColor           = "##ffffff"; }
  if( not isDefined( "attributes.sBorderColor" ))       { attributes.sBorderColor       = "##000000"; }
  if( not isDefined( "attributes.sAlignType" ))         { attributes.sAlignType         = "center"; }
  if( not isDefined( "attributes.sContentAlignType" ))  { attributes.sContentAlignType  = "left"; }
  if( not isDefined( "attributes.nContentPadding" ))    { attributes.nContentPadding    = "5px"; }
  if( not isDefined( "attributes.nBorderWidth" ))       { attributes.nBorderWidth = request.site.n_STANDARD_BOX_WIDTH; }
</cfscript>

<cfswitch expression="#thistag.executionMode#">
  <cfcase value="start">

    <cfif not thistag.hasEndTag>
      <cfoutput>
        <br/><br/>
        <strong><u>Error</u></strong>: [#cgi.script_name#]: missing endtag.
      </cfoutput>
      <cfexit method="exittag">
    </cfif>

    <cfoutput><div style="width:100%; padding:0px;"><div style="width:#attributes.nBorderWidth#; text-align:#attributes.sContentAlignType#; border:1px solid #attributes.sBorderColor#; padding:#attributes.nContentPadding#; background-color:#attributes.sBGColor#;"></cfoutput>
  </cfcase>

  <cfcase value="end">
    <cfoutput></div></div></cfoutput>
  </cfcase>
</cfswitch>

<cfsetting enablecfoutputonly="No">